UNITED STATES DEPARTMENT OF THE INTERIOR

GEOLOGICAL SURVEY

A low-cost digital recording and display system for Beckman UV 5200 series spectrophotometers

J.K. Crowley[1], D.M. Sherman[1], and I.J. Bennett[1]

Open-File Report 86 - 27

1986

[1]MS 927

National Center

Reston, Va 22092

Table of Contents

List of Figures

List of Tables

INTRODUCTION

For many research applications the ability to record data in digital format offers important advantages for data storage, display, and manipulation. Because of this, most new laboratory instruments offer digital recording as an option. However, it is also possible to retrofit older instruments to provide digital capability at substantially lower cost than the purchase of all new equipment. This paper describes a microcomputer-based digital recording system for use with Beckman Instruments, Inc., UV 5200 series spectrophotometers. The system utilizes readily obtainable components and requires no modifications to the spectrophotometer other than simple wiring connections. Total system cost is approximately $5,000.

SYSTEM HARDWARE DESCRIPTION

The heart of the digital recording system is an International Business Machines Corp., (IBM) Personal Computer equipped with a Scientific Solutions, Inc. Labmaster analog to digital (A/D) converter board. Other major components include a Hercules Computer Technology Graphics Card, which provides enhanced graphics resolution for displaying spectra, an IBM dot matrix printer, and an IBM X/Y 749 flatbed plotter. The individual equipment items that comprise the system are listed in table 1.

There are many ways to configure the A/D converter board for digital data recording, and a relatively simple arrangement is described here. The A/D sampling is initiated by a triggering pulse received from pin 4 on the TB401 Beckman spectrophotometer backplane connector (fig.1). The pulse is transmitted every 4 nanometers (nm) in the near-infrared portion of the spectrum (2600-700 nm) and every nanometer in the visible and ultra-violet wavelength range (700-300 nm). These pulse intervals set the upper limits on the system spectral resolution. Generation of the pulse requires that an optional Beckman wavelength marker accessory board be installed in the spectrophotometer.

Table 1, Equipment and Sources

1.	IBM PC or XT microcomputer - serial port - monochrome display - 5 1/4" disk drives (2) - 10 megabyte hard disk (optional) IBM X/Y 749 flatbed plotter IBM Graphics Printer IBM Basic Compiler	International Business Machines Corp. P.O. Box 1328-W Boca Raton, Florida 33432
2.	Hercules Graphics Card - parallel port - GraphX software	Hercules Computer Technology 2550 Ninth Street Suite 210 Berkeley, California 94710
3.	Labmaster Analog/Digital Converter - 12 bit resolution - connector cable	Scientific Solutions, Inc. 6225 Cochran Road Solon, Ohio 44139-3377
4.	Beckman Wavelength Marker Accessory Board	Beckman Instruments, Inc. 8920 Route 108 Columbia, Maryland 21045



Fig. 1. Summary of wiring connections: Spectrophotometer TB401 pin 4 (trigger pulse) is connected to the A/D mother board, J2 connector pin 3. Pins 1-2 and 4-20 on the J2 connector are not used. Pins 21-40 on the J2 connector are grounded to pin 15 on the TB401. Pin 14 on the TB401 (signal) is connected through a capacitor to the A/D daughter board P1D connector, pin 11. Pins 1-10 and 12-40 on the P1D connector are grounded back to pin 15 on the TB401. Pin 11 on the P1D connector is A/D channel 12.

Each pulse causes the A/D converter to sample a 0-1 volt analog signal, which is present on pin 14 of the TB401 connector (fig. 1). A 1 volt maximum signal corresponds to 100 percent transmission or reflectance, depending on the system operating mode. The A/D converter has 12 bit resolution and is designed to accept a 0-10 volt incoming signal. This signal is routed through a single A/D channel (channel 12 in listing 1, Appendix 1). If necessary, the 0-1 volt signal from the spectrophotometer can be amplified to fill more of the 12 bit A/D range, however, this will also amplify system noise. The graphics software presented here (listing 1, Appendix 1) assumes that the signal has been amplified to fill a 0-2 volt range. A substantial reduction in noise caused by the Beckman chopping motor may be obtained by placing a capacitor in the signal loop, as shown in figure 1.

Several additional considerations were involved in selecting system components. The flatbed-type plotter was selected to ensure optimum positioning accuracy for overlaying spectral data plots. The plotter also permits the user to interactively redefine the coordinate system origin using a button on the front panel. This feature is utilized by the software (listing 4, Appendix 1) to allow flexibility in the placement of labels and other text information during plotting.

The choice of graphics display card was determined by price and graphics resolution. The Hercules graphics card provides a resolution of 720 x 348 pixels, and also includes the Centronics-type parallel port needed to drive the dot matrix printer. In retrospect, the standard IBM graphics card (resolution 640 x 200 pixels) probably would have been adequate since the plotter is normally used for high-quality output. However, use of the standard IBM graphics card would require the separate purchase of a parallel port to operate the printer.

Two other items, although not required for the basic digital recording system, can significantly improve its overall performance. The first recommended item is a hard disk drive. Such drives provide increased mass storage capacity, as well as more rapid access to data files. The second recommended item is a Basic programming language compiler, which permits faster program execution.

SOFTWARE DESCRIPTION

The software developed for the digital recording system consists of three separate programs: 1) a program to record and display spectra, 2) a program that performs a 100 percent line correction, and other mathematical functions, and 3) a program for plotting spectral data. The three programs are listed in Appendix 1, and a help file describing the use of the software is included as Appendix II. This section focuses on design features and idiosyncrasies of the three programs.

The main program for recording and producing graphics displays of spectral data is entitled "SPECDATA" (listing 1, Appendix 1). Data can be recorded over any wavelength interval and a set of multiplicative correction factors that remove errors in the 100 percent line can be applied on a real-time basis. Stored data can also be redisplayed on the terminal screen, and hardcopies can be produced using the dot matrix printer.

SPECDATA is an unusual Basic program in that it utilizes special assembly language driver routines designed for the Hercules graphics display card. These routines are part of the GRAPHX software package available from Hercules Computer Technology, and are necessary in order to use the Hercules graphics card with the IBM Basic Compiler. While the driver routines greatly improve program speed and performance, they also introduce some peculiarities in the software coding. Chief among these is the necessity for a special subroutine to display text to the screen in graphics mode (program lines 1540 to 1620).

For comparison, a section of the SPECDATA program has been rewritten to permit data display on a standard IBM graphics display card (listing 2, Appendix 1). This version of the program only uses conventional Basic programming commands.

SPECCALC is a short program designed to permit simple numerical manipulations involving spectral data (listing 3, Appendix 1). Its primary uses are 1) to add/subtract spectra according to different weighting factors, 2) to correct for 100 percent line errors, and 3) to average multiple runs in order to improve signal to noise characteristics. The program can be expanded easily to perform other arithmetic functions.

SPECPLOT controls the IBM X/Y 749 flatbed plotter, and permits generation of high-quality hardcopy (listing 4, Appendix 1). It can accomodate most spectral data by providing plots in either of two formats, 1) visible and near-infrared combined, and 2) visible range only. The program permits the operator to interactively compose and place labels anywhere within the plotting area. The program also incorporates a scaling feature, which allows plots to be produced in different sizes, as well as a data smoothing algorithm to reduce the effects of system noise (program lines 5710-6910). A sample spectral data plot is shown in figure 2.

CONCLUSIONS

The digital data recording system discussed above has proven to be very serviceable, and is particularly appropriate for laboratories that have only limited electronics support. A chief advantage of the system is that no internal spectrophotometer modifications or complicated wiring connections are necessary. A disadvantage of the arrangement described is the inability to digitally integrate the signal over time in order to improve signal to noise characteristics. A more sophisticated hardware/software arrangement could combine signal integration with control of the spectrometer slitwidth to permit improved spectral resolution. With the present digital recording



Fig. 2. Sample output using IBM x/y 749 digital plotter.

system, it is possible, but not very efficient, to reduce noise by averaging multiple data runs. Fortunately, the data obtained in a single run are adequate for many spectral studies.

```
10 REM ************************************************************************
11 REM ********** SPECDATA Program to Acquire and Display Spectra **************
12 REM ************************************************************************
13 REM
20 CALL GMODE:CALL CLRSCR                'Starts graphics mode, clears screen
30 TX%=20:TY%=20:TX2%=500                'Sets coordinates for text messages
35 REM
40 DIM SPEC(1000):DIM WAVELEN(1000):DIM COR(1000)  'Declares arrays and
50 HI=2500:LO=400:IRVIS=800:SCREENFLAG=0:MODE=2    'sets defaults
60 REM
70 REM   ***** Program text messages *****
80 MSG0$="                                                                    "
90 MSG1$="Enter mode (1) acquire (2) display or (3) end                       "
100 MSG2$="Retain display?  y/n                                               "
110 MSG3$="Change scan parameters? y/n "
120 MSG4$="Enter filename                                                     "
130 MSG5$="Enter reflectance offset                                           "
140 MSG7$="Beginning of scan (nm)                                             "
150 MSG8$="End of scan (nm)                                                   "
160 MSG9$="IR-VIS breakpoint  (nm)                                            "
170 MSG10$="Strike any key when ready to start                                "
180 MSG11$="Switch to visible, hit any key when ready                         "
190 MSG12$="Enter spectrum label                                              "
200 MSG13$="Input 100% filename or `1' for no correction                      "
205 MSG14$="Scanning near-infrared wavelength range                           "
207 MSG15$="Scanning visible wavelength range                                 "
210 REM
220 REM   ***** Main Program *****
230 REM
240 CALL TEXTB(TX%,TY%,MSG1$):GOSUB 1540:N=VAL(A$) 'Enter mode selection
250 IF N=1 OR N=2 OR N=3 THEN MODE=N ELSE 260
260 ON MODE GOSUB 340,1040,1530
270 CALL TEXTB(TX%,TY%,MSG2$):GOSUB 1540:N$=A$     'Retain current display?
275 REM                                            'Default keeps screen
276 REM
280 IF N$="y" OR N$="Y" OR N$="" THEN SCREENFLAG=1:GOTO 240
290 IF N$="n" OR N$="N" THEN SCREENFLAG=0:GOTO 240
300 GOTO 270
310 REM
320 REM   ***** Subroutine to Acquire Spectral Data *****
330 REM
340 IF SCREENFLAG=0 THEN GOSUB 1300                'Redraws screen if needed
350 CALL TEXTB(TX%,TY%,MSG0$)                      'Clears top of screen
355 REM                                            'Next line displays scan
356 REM                                            'parameters
357 REM
360 PARAM$=STR$(HI)+" "+STR$(LO)+" "+STR$(IRVIS)+" "+MSG3$
361 REM
370 CALL TEXTB(TX%,TY%,PARAM$):GOSUB 1540:N$=A$    'Change scan parameters?
380 IF N$="y" OR N$="Y" THEN 390 ELSE 440
390 CALL TEXTB(TX%,TY%,MSG0$)
400 CALL TEXTB(TX%,TY%,MSG7$):GOSUB 1540:HI=VAL(A$) 'Get new parameters
410 CALL TEXTB(TX%,TY%,MSG8$):GOSUB 1540:LO=VAL(A$)
420 CALL TEXTB(TX%,TY%,MSG9$):GOSUB 1540:IRVIS=VAL(A$)
```

```
430 GOTO 350
440 BRK=INT((HI-IRVIS)/4+1):NUM=BRK+IRVIS-LO-1      'Calculate VIS/IR breakpoint
441 REM                                             'and total number of points
450 CALL TEXTB(TX%,TY%,MSG13$):GOSUB 1540:COR$=A$   'Input 100% correction file
470 IF COR$="" THEN 680                             'Uses current correction
480 IF COR$="1" THEN 600                            'or no correction
490 OPEN COR$ FOR INPUT AS #1
500 INPUT #1, V$                                    'Program version number
510 INPUT #1, LAB$                                  'Correction file label
520 INPUT #1, HI2 ,LO2,IRVIS2                       'Correction file header
525 IF HI2<HI THEN J=(HI2-HI)/4                     'Set index to match the
526 REM                                             'correction file with data
527 REM                                             'being recorded
528 REM
530 BRK2=INT((HI2-IRVIS2)/4+1):NUM2=BRK2+IRVIS2-LO2-1  'New breakpoint/total
540 FOR I=0 TO NUM2
550 INPUT #1,XCOR, COR(I)                           'Read wavelength/reflectance
560 IF XCOR=HI THEN J=1                             'In some cases the data and
561 REM                                             'the correction file may
562 REM                                             'start at same wavelength
570 NEXT I
580 CLOSE
590 GOTO 630
600 FOR I=0 TO NUM
610 COR(I)=1                                        'No correction performed
620 NEXT I
630 REM                                             'Initialize array of
640 FOR I=0 TO NUM                                  'wavelength values
650 IF I<BRK THEN WAVELEN(I)=HI-(I*4)               'IR increment 4 nm
660 IF I=>BRK THEN WAVELEN(I)=IRVIS+BRK-I-1         'VIS increment 1 nm
670 NEXT I
680 CALL TEXTB(TX%,TY%,MSG0$)
690 CALL TEXTB(TX%,TY%,MSG10$)                      'Strike any key when ready
700 IF INKEY$="" THEN 700
705 IF HI>800 THEN CALL TEXTB(TX%,TY%,MSG14$) ELSE CALL TEXTB(TX%,TY%,MSG15$)
706 REM
707 REM    ***** Perform A/D Conversion *****
708 REM
710 ADDRESS=1808:CHANNEL=12                         'Initializes I/O address
720 OUT ADDRESS+5,CHANNEL                           'and A/D channel number
730 FOR I=0 TO NUM
735 IF INKEY$="i" THEN 270                          'Permits scan interruption
740 OUT ADDRESS+4,132                               'Starts a conversion
750 IF INP(ADDRESS+4)<128 THEN 750                  'Waits for conversion finish
760 LOSEG=INP(ADDRESS+5):HISEG=INP(ADDRESS+6)       'Reads data
770 Y=256*HISEG+LOSEG
772 IF I+J<0 THEN SPEC(I)=Y:GOTO 807                'No correction if data range
773 IF COR(I+J)=0 THEN SPEC(I)=Y:GOTO 807           'is outside the correction
775 SPEC(I)=INT(Y/COR(I+J)+.5)                      'file range
776 REM
780 IF I=BRK-1 THEN CALL TEXTB(TX%,TY%,MSG11$) ELSE GOTO 807 'Checks for VIS/IR
790 BEEP                                                     'breakpoint
800 IF INKEY$="" THEN 800
805 CALL TEXTB(TX%,TY%,MSG15$)                      'Indicate start of VIS range
```

```
807 WAVELEN=WAVELEN(I)
808 IF WAVELEN(I) >2600 THEN WAVELEN=WAVELEN(I)-200          'Shifts data for
810 IF WAVELEN(I)<400 THEN WAVELEN=WAVELEN(I)+200           'display purposes
811 REM
820 X%=(WAVELEN*.32)-120                                    'Scales data to Hercules car
830 Y%=(SPEC(I)*-.3907)+346                                 'Next line suppresses spikes
835 REM
840 IF I<2 OR I=BRK+2 OR I=BRK+1 THEN CALL MOVE(X%,Y%) ELSE CALL DLINE(X%,Y%)
860 NEXT I
865 REM
870 REM   ***** Output file routine *****
880 BEEP
890 CALL TEXTB(TX%,TY%,MSG4$):GOSUB 1540:SPECNM$=A$         'Enter filename
900 TX2%=250
910 CALL TEXTB(TX%,TY%,MSG12$):GOSUB 1540:LAB$=A$           'Enter label
920 TX2%=500
930 LAB$=SPECNM$+"   "+LAB$
940 OPEN SPECNM$ FOR OUTPUT AS #1
950 PRINT #1,"V3" '                                         'Program version number
960 PRINT #1, LAB$                                          'Spectrum label
970 PRINT #1, HI,LO,IRVIS '                                 'Header information
980 FOR I=0 TO NUM
990 REFL=SPEC(I)/819                                        'Convert to reflectance
1000 PRINT #1, USING "#### #.###":WAVELEN(I),REFL
1010 NEXT I
1020 CLOSE:RETURN
1030 REM
1040 REM   ***** Subroutine to Display Spectral Data ******
1050 REM
1060 CALL TEXTB(TX%,TY%,MSG0$)
1070 CALL TEXTB(TX%,TY%,MSG4$)                              'Enter filename for display
1080 GOSUB 1540:SPECNM$=A$
1090 CALL TEXTB(TX%,TY%,MSG5$)                              'Enter reflectance offset
1100 GOSUB 1540:OFFSET=VAL(A$)
1110 IF SCREENFLAG=0 THEN GOSUB 1300                        'Redraw screen if needed
1120 OFFSET=OFFSET*(YL%-YH%)/100
1130 OPEN SPECNM$ FOR INPUT AS #1
1140 ON ERROR GOTO 1630                                     'Trap filename errors
1150 INPUT #1, V$
1160 INPUT #1, LAB$
1170 CALL TEXTB(TX%,TY%,LAB$)
1180 INPUT#1, HI,LO,IRVIS
1190 BRK=INT((HI-IRVIS)/4+1):NUM=BRK+IRVIS-LO-1             'Calculate breakpoint/total
1200 FOR I=0 TO NUM                                         'points in data
1210 INPUT #1,WAVELEN,REFL
1220 IF WAVELEN<400 THEN WAVELEN=WAVELEN+200                'Shift data for display if
1230 X%=(WAVELEN*.32)-120                                   'necessary
1240 Y%=(819*REFL*-.3907)+346+OFFSET                        'Scale to Hercules card
1241 REM
1250 IF I<2 OR I=BRK+2 OR I=BRK+1 THEN CALL MOVE(X%,Y%) ELSE CALL DLINE(X%,Y%)
1260 NEXT I
1270 T$=INKEY$:IF INKEY$="" THEN 1270                       'Pause with label on screen
1280 CLOSE:RETURN                                           'until a key is struck
1290 REM
```

```
1300 REM    ***** Subroutine to Draw Screen Using GraphX Functions *****
1310 REM
1320 SCREENFLAG=1
1330 CALL CLRSCR
1340 XL%=8:XH%=712:YL%=26:YH%=346                    'Sets screen corner values
1350 CALL MOVE(XL%,YL%)                              'Draws sides of box
1360 CALL DLINE(XH%,YL%):CALL DLINE(XH%,YH%)
1370 CALL DLINE(XL%,YH%):CALL DLINE(XL%,YL%)
1380 XR%=(XH%-XL%)/22:XS%=(XH%-XL%)-(2*XR%):YR%=(YH%-YL%)/10
1381 REM
1390 FOR X%=XL% TO XH% STEP XR%                      'Draws vertical dashed lines
1400 FOR Y%=YL% TO YH% STEP YR%/3
1410 CALL MOVE(X%,Y%):Y%=Y%+2:CALL DLINE(X%,Y%)
1420 NEXT Y%
1430 NEXT X%
1440 FOR Y%=YL% TO YH% STEP YR%
1450 CALL MOVE(XL%,Y%):X%=XL%+4:CALL DLINE(X%,Y%)   'Draws reflectance tickmarks
1460 CALL MOVE(XH%,Y%):X%=XH%-4:CALL DLINE(X%,Y%)
1470 NEXT Y%
1480 FOR X%=XL%+XR% TO XH%-XR% STEP XS%/4
1490 Y%=YL%-4:CALL MOVE(X%,Y%):CALL DLINE(X%,YL%)
1500 Y%=YH%-4:CALL MOVE(X%,Y%):CALL DLINE(X%,YH%)
1510 NEXT X%
1520 RETURN
1530 CALL TMODE:END
1531 REM
1540 REM    ***** Subroutine to Send Text to Screen *****
1541 REM
1550 A$="":C$=""
1560 C$=INPUT$(1)
1570 IF ASC(C$)=13 THEN RETURN                       'Checks for a RETURN keystroke
1580 A$=A$+C$
1590 IF ASC(C$)=8 THEN A$=LEFT$(A$,LEN(A$)-2)       'Checks for a backspace
1600 B$=A$+" "
1610 CALL TEXTB(TX2%,TY%,B$)
1620 GOTO 1560
1630 IF (ERR=53) THEN RESUME 1060
```

```
1 REM ************************************************************************
2 REM ***** DISPLAY Program to Display Spectra on an IBM Graphics Card ******
3 REM ************************************************************************
10 SCREEN 2:CLS
20 DIM SPEC(1000):DIM WAVELEN(1000)
30 HI=2500:LO=400:IRVIS=800:SCREENFLAG=0
40 GOSUB 320
70 LOCATE 1,1:INPUT "Enter filename for display";SPECNM$
80 LOCATE 1,1:INPUT "Enter reflectance offset (default=0)";OFFSET$
90 IF OFFSET$="" THEN OFFSET=0 ELSE OFFSET=VAL(OFFSET$)
100 OFFSET=OFFSET*(YL-YH)/100
110 OPEN SPECNM$ FOR INPUT AS #1
115 ON ERROR GOTO 520
116 INPUT #1,V$                                    'Program version number
117 INPUT #1, LAB$                                 'Spectrum label
120 INPUT #1, HI,LO,IRVIS                          'Spectrum header
150 BRK=INT((HI-IRVIS)/4+1):NUM=BRK+IRVIS-LO-1     'Calculate # of data points
200 FOR I=0 TO NUM
210 INPUT #1, WAVELEN,REFL                         'Read wavelength/reflectance
215 IF WAVELEN<400 THEN WAVELEN=WAVELEN+200        'Shifts short wavelength data
220 X=(WAVELEN*.2836)-105.45                       'for viewing purposes only
230 Y=819*REFL*(-.2198)+198+OFFSET                 'Next line suppresses spikes
235 REM
240 IF I<2 OR I=BRK+2 OR I=BRK +1 THEN PRESET (X,Y) ELSE LINE -(X,Y)
250 NEXT I
275 LOCATE 1,1:PRINT"                                                          "
280 LOCATE 1,1:PRINT LAB$
290 T$=INKEY$:IF INKEY$="" THEN 290                'Pause with label on screen
300 SCREENFLAG=1                                   'until a key is struck
310 CLOSE
315 LOCATE 1,1:INPUT"Clear display? Enter y/n or q to quit";A$
317 IF A$="y" OR A$="Y" THEN GOSUB 320
318 IF A$="q" THEN SCREEN 0:END
319 GOTO 70
320 REM ***** Subroutine to draw grid *****
340 CLS:XL=8:XH=632:YL=18:YH=198                   'Sets grid corners
345 REM
350 PRESET(XL,YL):LINE -(XH,YL):LINE -(XH,YH):LINE -(XL,YH):LINE -(XL,YL)
360 XR=(XH-XL)/22:XS=(XH-XL)-(2*XR):YR=(YH-YL)/10
370 FOR X=XL TO XH STEP XR
380 FOR Y=YL TO YH STEP YR/3
390 PRESET(X,Y):Y=Y+2:LINE -(X,Y)
400 NEXT Y
410 NEXT X
420 FOR Y=YL TO YH STEP YR
430 PRESET(XL,Y):X=XL+4:LINE -(X,Y)
440 PRESET(XH,Y):X=XH-4:LINE -(X,Y)
450 NEXT Y
460 FOR X=XL+XR TO XH-XR STEP XS/4
470 Y=YL-4:PRESET(X,Y):LINE -(X,YL)
480 Y=YH-4:PRESET(X,Y):LINE -(X,YH)
490 NEXT X
500 RETURN
520 IF (ERR=53) THEN RESUME 70
```

```
1 REM ************************************************************************
2 REM ********** SPECCALC Program for Mathematical Operations ***************
3 REM ************************************************************************
4 REM
10 KEY OFF:CLS:SF=1:FLAG=0:J=0
20 DIM TOT(1000):DIM WAVELEN(1000):DIM CURRENT(1000)
30 REM
40 REM ****** Main program *******
41 REM
50 INPUT"Enter filename for calculation ",NF$
55 IF NF$="" THEN CLOSE:END
60 INPUT"Enter scale factor (default=1) ",S$
70 IF S$><"" THEN SF=VAL(S$)
80 GOSUB 180
85 PRINT"       "
90 PRINT"Enter 'o' to output current data in accumulator"
100 PRINT "Enter 'c' to perform 100% correction on data in accumulator"
110 PRINT "Enter + or - to add/subtract next file from data in accumulator"
120 INPUT ANS$
125 PRINT:PRINT
130 IF ANS$="o" THEN GOSUB 300
140 IF ANS$="c" THEN INPUT"Enter 100% filename ",NF$:FLAG=1:GOSUB 180
150 IF ANS$="+" THEN FLAG=0
160 IF ANS$="-" THEN FLAG=2
170 GOTO 50
171 REM
180 REM ***** Input file routine ******
181 REM
190 OPEN NF$ FOR INPUT AS #1
195 ON ERROR GOTO 430                                   'Traps filename entry errors
197 INPUT #1, V$                                        'Program version number
198 INPUT #1,LAB$                                       'Data file label
199 REM                                                 'Next line saves data header
200 REM                                                 'so that 100% file can be read
201 REM
202 IF FLAG=1 THEN SAVEHI=HI:SAVELO=LO:SAVEIRVIS=IRVIS:SAVENUM=NUM
210 INPUT #1,HI,LO,IRVIS
214 REM
215 IF FLAG=1 AND HI < SAVEHI THEN J=(HI-SAVEHI)/4    'Allows use of 100% line
216 REM                                                 'file of different size
217 REM                                                 'than the data file.
218 REM
220 BRK=INT((HI-IRVIS)/4+1):NUM=BRK+IRVIS-LO-1  'Calculates total points in
225 REM                                          data and the breakpoint
226 REM                                          between the IR and VIS.
230 FOR I=0 TO NUM
240 INPUT #1,WAVELEN(I),CURRENT(I)                      'Input wavelength and reflec-
250 IF FLAG=0 THEN TOT(I)=TOT(I)+CURRENT(I)*SF  'tance, and scale data as it
260 IF FLAG=1 AND WAVELEN(I)=SAVEHI THEN J=I    'is placed in the accumulator.
270 IF FLAG=2 THEN TOT(I)=TOT(I)-CURRENT(I)*SF  'Flags are used to identify the
280 NEXT I                                              'appropriate math operation.
281 REM
290 CLOSE #1:IF FLAG=1 THEN GOSUB 300 ELSE RETURN     'If 100% correction is
```

```
291 REM                                                    'being performed goto
292 REM                                                    'output file routine.
300 REM ****** Output file routine *******
301 REM
302 REM                                                 'Next line resets header info
303 REM                                                 'from values used to process
304 REM                                                 '100% file to values used for
305 REM                                                 'the data file
306 REM
307 IF FLAG=1 THEN HI=SAVEHI:LO=SAVELO:IRVIS=SAVEIRVIS:NUM=SAVENUM
310 INPUT"Enter output filename ",SPECNM$
320 INPUT"Enter label for output file ",LAB$
325 IF LAB$="" THEN LAB$=" "                           'Label can't be a nul character
330 OPEN SPECNM$ FOR OUTPUT AS #1
335 PRINT #1,"V3"
336 PRINT #1, LAB$
340 PRINT #1,HI,LO,IRVIS
350 FOR I=0 TO NUM
354 IF FLAG=1 AND TOT(I)=CURRENT(I+J) THEN GOTO 370   'The 100% correction is
356 IF FLAG=1 AND I+J<0 THEN GOTO 370                  'only performed where the
360 IF FLAG=1 THEN TOT(I)=TOT(I)/CURRENT(I+J)          '100% file overlaps the
370 PRINT #1,USING"#### #.###";WAVELEN(I+J),TOT(I)     'data file.
410 NEXT I
420 CLOSE
425 RETURN
430 IF (ERR=53) THEN RESUME 50
```

```
10 REM ************************************************************************
15 REM **** SPECPLOT Program to Plot Spectra Using an IBM X/Y 749 Plotter ****
16 REM ************************************************************************
17 REM
20 KEY OFF:CLS:XO=200:Y)=200
110 RESET:FLAG=0:FLAG2=0
210 PRINT"Turn on plotter, load chart, place pens in slots #1 and #2"
310 PRINT"and press `plot' button"
320 REM
410 OPEN "COM1:2400,N,8,1,CS65535" AS #1      'Open communications port
420 REM
510 PRINT:PRINT"Enter filename for first plot":INPUT NF$
610 REM
710 REM **** OPEN SPECTRA DATA FILE AND READ HEADER *****
810 REM
910 DIM X(1000),Y(1000), SY(1000)
1005 ON ERROR GOTO 15115                      'Traps filename entry errors
1010 OPEN NF$ FOR INPUT AS #2
1015 INPUT #2,V$                              'Program version number
1016 INPUT #2,LAB$                            'Spectrum label
1110 INPUT #2, HI,LO,IRVIS                    'Gets spectrum header infomation
1210 REM
1310 REM **** SET UP PLOTTING WINDOW AND DRAW AXES *****
1410 REM
1510 PRINT:PRINT"Enter scale factor from 0.3 to 1.0":INPUT S
1610 S=1/S:P$="H"
1810 TOTVIS=IRVIS-LO:TOTIR=HI-IRVIS:TOTNM=HI-LO   'Determines number of points
1910 IF TOTIR=-1 THEN TOTNM=TOTNM*3:FLAG2=1       'in each segment and type
1911 REM                                          'of plot.
1912 REM
2010 PRINT"Enter vertical offset in +/- reflectance":INPUT VO
2110 OFFSET=(1638/S)*VO/100
2115 PRINT "Do you want an absolute reflectance scale?  (y/n)":INPUT ANS$
2120 PRINT "Do you want vertical gridlines? (y/n)":INPUT GRID$
2210 IF FLAG=1 THEN PRINT #1,"T 4";XO;YO:GOTO 4510
2310 REM                                          'Origin may be redefined
2311 REM                                          'by editing line 20
2312 REM
2410 PRINT #1, "F 1"                          'Get pen 1  for axes plot
2510 PRINT #1, "F 10,16"                      'Slow pen to appropriate speed
2610 PRINT #1, "T 4";XO;YO                    'Define origin

2710 PRINT #1, XO;"/";YO;"HK"                 'Go to the origin to begin axes
2810 PRINT #1, TOTNM/S + XO;"/";YO;"IK"       'Draw lower x axis
2910 PRINT #1, "Y";1638/S;164/S;0;-20/S       'Draw y axis on right
3010 PRINT #1, XO;"/";YO;"HK"                 'Go back to the origin
3110 PRINT #1, "Y";1638/S:164/S:0;20/S        'Draw y axis on left
3210 PRINT #1, TOTNM/S + XO;"/";YO + 1638/S;"IK"
3310 IF FLAG2=1 THEN LOC1=450 ELSE LOC1=1000  'Sets label positions
3410 IF FLAG2=1 THEN LOC2=300 ELSE LOC2=875   'depending on plot type
3510 PRINT #1, 0+S*30;"/";800/S:"HK"
3610 PRINT #1, "Z";60/S:90;60/S;"BREFLECTANCE"     'Reflectance label
3710 PRINT #1, LOC1/S;"/";0+S*30;"HK"
3810 PRINT #1, "Z";60/S;0;60/S;"BWAVELENGTH (UM)"  'Wavelength label
3815 IF GRID$="y" THEN 4110
```

```
3910 PRINT #1, LOC2/S;"/";1.5*YO + 1638/S;"HK"
4010 PRINT #1, "Z";60/S;0;60/S;"BWAVENUMBER (1000/CM)"
4110 GOSUB 12710
4210 GOSUB 9510
4510 PRINT #1, "F 2" :                          'Get pen #2 for spectral plot
4610 REM
4710 REM ***** READ DATA FROM FILE AND CONVERT TO PLOTTER COORDINATES ******
4810 REM
4910 TOTIR=TOTIR/4+1                            'Calculates number of IR points
5010 FOR I=0 TO TOTVIS+TOTIR-1                  'Loop over total number of points
5110 INPUT #2, WAVELEN,Y(I)                     'Input wavelength and reflectance
5210 Y(I)=Y(I)*1638/S + YO + OFFSET             'Scale and offset reflectance
5212 REM                                        'Next line suppresses spikes
5213 REM                                        'commonly seen at start of data
5214 REM
5215 IF I=1 OR I=TOTIR+2 OR I=TOTIR+1 THEN Y(I-1)=Y(I)
5310 X(I)=(WAVELEN-LO)/S
5510 NEXT I
5610 REM
5710 REM ***** SMOOTH DATA USING BINOMIAL EXPANSION ******
5810 REM
5910 FOR J=3 TO (TOTVIS + TOTIR-2)              'Smooth over 3 channels in IR
6010 REM
6110 SY(J)=Y(J-1)/4 + Y(J)/2 + Y(J+1)/4
6210 REM                                        'Smooth over 5 channels in VIS
6310 IF J>TOTIR+2 THEN SY(J)=Y(J-2)/10 +Y(J-1)/5 + 2*Y(J)/5 + Y(J+1)/5 +
Y(J+2)/10
6610 IF FLAG2=1 THEN X(J)=X(J)*3
6710 PRINT #1, X(J)+XO;"/";SY(J);P$;"K"
6810 P$="I"
6910 NEXT J
7010 PRINT #1, "F"                              'Replace used pen
7110 PRINT #1, XO;"/";YO;"HK"
7210 CLOSE #2
7310 REM
7410 REM ***** LABEL SPECTRUM AND/OR READ A NEW SPECTRUM FILE ******
7510 REM
7610 PRINT:PRINT"Change pen #2 and enter next filename, or press RETURN to label
spectra"
7710 INPUT NF$:IF NF$="" THEN 7810 ELSE FLAG=1:GOTO 1010
7810 PRINT:PRINT"Enter label for spectrum"
7910 LINE INPUT LAB$
8010 PRINT"1) End plot mode by pressing the plot button."
8110 PRINT"2) Move the pen using the cursor keys.":PRINT"3) Press SELECT ZERO
when pen holder is positioned."
8210 PRINT"4) Ensure that pen #2 color matches spectrum being labeled.":PRINT"5)
Return to plot mode."
8310 PRINT:PRINT"Hit any key when ready to proceed."
8410 IF INKEY$="" THEN 8410
8510 PRINT #1, "F 2"
8610 PRINT #1, "0/0HK"                          'Interactively defines the

8710 PRINT #1, "Z";40/S;0;40/S;"B";LAB$         'label position
8810 PRINT #1, "F"
8910 PRINT "Any more labels?  (y/n)"
```

```
9010 INPUT ANS$:IF ANS$="y" THEN 7810 ELSE PRINT"PLEASE CAP AND REPLACE PENS"
9110 CLOSE
9210 END
9410 REM
9510 REM ***** SUBROUTINE TO LABEL AXES ******
9610 REM
9710 IF FLAG2=1 THEN GOTO 11710                'Checks for visible only plot
9810 XA = XO +(450-LO)/S
9910 PRINT #1, XA;"/";YO - 70/S;"HK"
10010 IF TOTVIS<=1 THEN GOTO 10210
10110 PRINT #1, "Z";40/S;0;40/S;"B0.5"
10210 PRINT #1, XA+500/S;"/";YO - 70/S;"HK"
10310 PRINT #1,"Z";40/S;0;40/S;"B1.0"
10410 PRINT #1, XA+1000/S;"/"YO - 70/S;"HK"
10510 PRINT #1,"Z";40/S;0;40/S;"B1.5"
10610 PRINT #1, XA+1500/S;"/"YO - 70/S;"HK"
10710 PRINT #1, "Z";40/S;0;40/S;"B2.0"
10810 PRINT #1, XA+2000/S:"/"YO - 70/S;"HK"
10910 PRINT #1, "Z";40/S;0;40/S;"B2.5"
10920 IF GRID$="y" THEN GOTO 11910
11010 XA = XO + (475-LO)/S
11110 PRINT #1, XA;"/";YO + 1668/S;"HK"
11210 PRINT #1, "Z";40/S;0;40/S;"B20"
11310 PRINT #1, XA+490/S;"/";YO + 1668/S;"HK"
11410 PRINT #1, "Z";40/S;0;40/S;"B10"
11510 PRINT #1, XA+1515/S;"/";YO + 1668/S:"HK"
11610 PRINT #1, "Z";40/S;0;40/S;"B5"
11615 GOTO 11910
11710 REM                                      'Labels for visible only
11715 XA=XO+250/S
11720 PRINT #1, XA;"/";YO-70/S;"hk"
11725 PRINT #1,"z";40/S;0;40/S;"b0.5"
11730 PRINT #1, XA+300/S;"/";YO-70/S;"hk"
11735 PRINT #1, "z";40/S;0;40/S;"b0.6"
11740 PRINT #1, XA+600/S;"/";YO-70/S;"hk"
11745 PRINT #1, "z";40/S;0;40/S;"b0.7"
11750 PRINT #1, XA+900/S;"/";YO-70/S;"hk"
11755 PRINT #1, "z";40/S;0;40/S;"b0.8"
11757 IF GRID$="y" THEN 11910
11760 PRINT #1, XA+20/S:"/";YO+1668/S;"hk"
11765 PRINT #1, "z":40/S;0;40/S;"b20"
11770 PRINT #1, XA+515/S;"/";YO+1668/S;"hk"
11775 PRINT #1, "z";40/S;0;40/S;"b15"
11910 IF ANS$="n" THEN RETURN
12010 PRINT #1, XO-50/S;"/";YO-20/S;"HK"
12110 PRINT #1,"Z";40/S;0;40/S;"B0"
12210 PRINT #1, XO-90/S;"/";YO+800/S;"HK"
12310 PRINT #1, "Z";40/S;0;40/S;"B50"
12410 PRINT #1, XO-130/S:"/";YO+1620/S;"HK"
12510 PRINT #1, "Z";40/S;0;40/S;"B100"
12610 RETURN
12710 REM ****** SUBROUTINE TO DRAW TICKMARKS ON WAVELENGTH AXIS ******
12711 REM
12810 PRINT #1, XO:"/";YO;"HK"
```

```
12910 XA = XO + (400-LO)/S
13010 IF FLAG2=0 THEN LAST =INT(TOTNM/100)+1 ELSE LAST=INT(TOTNM/300)+1
13110 FOR K = 1 TO LAST
13210 PRINT #1, XA;"/";YO;"HK"
13310 IF (K MOD 5) = 2 THEN YT = YO + 40/S ELSE YT = YO + 20/S
13410 PRINT #1, XA;"/";YT;"IK"
13415 IF GRID$<>"y" THEN 13510
13417 IF K=1 OR K=LAST THEN 13510
13420 FOR I=58/S TO 1638/S STEP 120/S
13430 PRINT #1, XA;"/";YO+I;"hk"
13435 PRINT #1, XA;"/";YO+I+10/S;"ik"
13440 NEXT I
13510 IF FLAG2=0 THEN XA = XA + 100/S
13610 IF FLAG2=1 THEN XA=XA+300/S
13710 NEXT K
13720 IF GRID$="y" THEN RETURN
13721 REM
13810 REM ***** PUT TICK MARKS ON WAVENUMBER SCALE *****
13910 CM = 25
14010 YA = YO + 1638/S
14110 IF FLAG2=1 THEN LAST=13
14210 FOR K = 1 TO LAST
14310 XA = XO + (10000/CM - LO)/S
14410 IF FLAG2=1 THEN XA=XO+(10000/CM-LO)*3/S
14510 IF XA > (XO + TOTNM/S) GOTO 15010
14610 PRINT #1, XA;"/";YA;"HK"
14710 IF (K MOD 5) = 1 THEN YT = (YA-40/S) ELSE YT = (YA-20/S)
14810 PRINT #1, XA;"/";YT;"IK"
14910 CM = CM - 1
15010 NEXT K
15110 RETURN
15115 IF (ERR=53) THEN RESUME 510
```

Appendix II

General Instructions for Use of SPEC Software

1) Insert "Spec Software" diskette in drive A, data diskette in drive B. The master diskette should contain the following files:

Command.com	(IBM)	Int10.com (Hercules)
Format.com	(IBM)	Specdata.exe
Basrun.exe	(IBM)	Speccalc.exe
Basrun.lib	(IBM)	Specplot.exe
		100%.fil

Execute the INT10.COM to initialize graphics routines then proceed.

2) There are three main programs:

SPECDATA- Records and displays spectral data

SPECCALC- Permits averaging and correcting spectra

SPECPLOT- Produces smoothed graphics plots of spectra

3) The operator must have a formatted data diskette. To make a formatted diskette type FORMAT B: (RETURN) and follow the instructions. Any data already on the diskette being formatted will be lost!

4) SPECDATA- To use this program simply type "specdata", followed by RETURN.

a) Select either the record or display option by entering the appropriate number from the menu.

b) This program allows data to be recorded over any wavelength range. Sample scan parameters:

VISIBLE ONLY	VISIBLE/IR
scan start=800	scan start=2500
scan end =400	scan end = 400
breakpoint=801	breakpoint= 800

Answer the "CHANGE SCAN PARAMETERS Y/N" prompt with "n" RETURN, when the parameters are correct.

c) The scan offset is normally 0. Different offsets may be used to reduce overlap when displaying several spectra simultaneously. Offsets are in percent reflectance and may be positive or negative.

d) The program can either perform the 100 percent line correction on a real-time basis as the data are recorded, or the correction can be performed later using SPECCALC. To correct real-time, the operator must supply the name of an existing file that contains an average of 10-20 100% line spectra. (The more spectra in this average, the better the noise reduction). Once the correction file has been entered at the start of a data recording session, the file will continue in use, provided that the SPECDATA program does not fail and have to be restarted. Press RETURN in response to the correction filename prompt during subsequent runs. To record data without any correction, for example when recording 100% lines, enter "1" in response to the correction filename prompt. This will fill the correction factor array with ones. Again, the operator may simply hit RETURN in response to the prompt on subsequent runs.

e) After the correction array is loaded, press any key to initialize the computer to receive data. All spectrometer switches should be set to their correct positions before the computer is initialized. This will help to minimize spurious data at the beginning of a spectrum.

f) After a data recording run is completed, the operator is asked for an output filename. This filename can be up to eight characters long, plus a three character extension. No blanks or commas are allowed.

g) Labels can be up to 60 characters long; blanks are allowed, but commas are not.

h) Generally, it is only necessary to enter a response to a prompt once. For example, entering a "2" at the first prompt will place the program in display mode. The program will remain in this mode until the operator changes it. Hit RETURN at the prompt to continue displaying spectra.

i) To interrupt a scan, press "i" while the scan is in progress.

5) INSTRUCTIONS FOR SPECCALC:

a) This program is used to average spectra, and to add and subtract combinations of spectra. Each spectrum is scaled and entered into an "accumulator". For example, to average three spectra, each file is scaled by .333; To mix 20% of A plus 80% of B, scale the filename entries by 0.2 and 0.8, respectively.

b) Each spectrum is entered and scaled, and the operator is asked to choose an action. The choices are:

-Add the next spectrum to the data in the accumulator.

-Subtract the next spectrum from the data in the accumulator

-Correct the data in the accumulator using a 100% file

-Output the data in the accumulator-obviously the last step once the other steps are finished.

6) INSTRUCTIONS FOR USING SPECPLOT:

a) This program is almost self-explanatory; however, the operator must pay close attention to all program instructions. The program automatically performs a binomial smoothing routine that helps to minimize any residual noise in the recorded data.

b) The program can produce plots of various sizes according to a scale factor. Recommended scale factors are: 1.0 (maximum size), 0.8 , 0.6 , and 0.4. The smaller factors give thicker relative line widths. Transparency film and high gloss paper are available for making final plots. Plain white typing paper can be used for test plots.

c) Labels can be placed anywhere on the plot using the labeling routine. Follow the instructions in the program carefully. Note that the SELECT ZERO buttons should be pressed sequentially, not simultaneously.